CAPITOLO 9
GLI ALGORITMI DI CORRELAZIONE

Vanno sotto il nome di algoritmi di correlazione una serie di strumenti matematici che eseguono la ricerca dei legami di interdipendenza tra fenomeni che interessano tutte le scienze.
La possibilità di implementare in Qbasic questi algoritmi offre al lettore un efficace mezzo di indagine in molti campi degli studi e della tecnica.

9.1 La correlazione tra grandezze in numero discreto

Correlazione è la dipendenza reciproca tra due serie di grandezze, l'entità della dipendenza è definita come **coefficiente di correlazione** (C). Se le due serie di grandezze sono strettamente dipendenti l'una dall'altra si ha un elevato coefficiente di correlazione positivo o negativo, se le due serie di grandezze sono poco dipendenti tra loro si ha un modesto valore del coefficiente di correlazione, se infine le due serie di grandezze sono totalmente indipendenti si ha un coefficiente di correlazione nullo.

Il massimo valore del coefficiente di correlazione è dato da $C = 1$ o da $C = -1$
Il minimo valore del coefficiente di correlazione è dato da $C = 0$

Chiariamo la cosa prendendo due serie di grandezze individuate rispettivamente dalle lettere:

A1; A2; ....; An   e   B1; B2;.....; Bn

calcoliamo le **medie** delle due serie, Am e Bm:

$$Am = (A1 + A2 +...+ An) / n \qquad Bm = (B1 + B2 +...+ Bn) / n$$

impostiamo il rapporto:

$$C = \frac{(A1-Am) (B1-Bm) + (A2-Am) (B2-Bm) + ...+ (An-Am) (Bn-Bm)}{\left[[(A1-Am)^2 +(A2-Am)^2 + .. +(An-Am)^2] \; [(B1-Bm)^2 +(B2-Bm)^2 +..+(Bn-Bm)^2] \right]^{1/2}}$$

il valore di C esprime il coefficiente di correlazione tra le due serie di valori A1...An; B1;...Bn.
Nei processi statistici sono significativi inoltre i valori detti di **deviazione standard** che per ciascuna delle due serie sono calcolabili con le seguenti espressioni:

$$DsA = \left[\left((A1-Am)^2 + (A2-Am)^2 + ... + (An-Am)^2 \right) / n \right]^{1/2}$$

$$DsB = \left[\left((B1-Bm)^2 + (B2-Bm)^2 + ... + (Bn-Bm)^2 \right) / n \right]^{1/2}$$

Per rendere più tangibile il significato del coefficiente di correlazione è necessario implementare in un programma in Qbasic gli algoritmi che abbiamo ora mostrato ed operare con questi su

particolari serie di valori che possano illustrare come le caratteristiche del loro legame siano esprimibili mediante il valore di C.
Il programma consente il calcolo, sia del valore del coefficiente di correlazione C di cui ci occuperemo, sia dei valori medi Am; Bm e delle deviazioni standard DsA; DsB che possono essere utili a chi si interessa di problemi statistici.
La compilazione del programma richiede una scomposizione delle formule date in modo che risulti più facile trovare le corrispondenze simboliche in linguaggio Qbasic; il commento al programma sotto elencato mette in evidenza questa scomposizione che si avvale delle seguenti memorie di appoggio:

A = sommatoria progressiva dei termini An per il calcolo delle medie Am
B = sommatoria progressiva dei termini Bn per il calcolo delle medie Bm
Ao = differenze alle medie per i termini An
Bo = differenze alle medie per i termini Bn
SP = sommatoria progressiva dei prodotti delle sommatorie
Aq = sommatoria progressiva dei quadrati delle differenze alla media per la serie An
Bq = sommatoria progressiva dei quadrati delle differenze alla media per la serie Bn
RAq = calcolo radice quadrata di Aq
RBq = calcolo radice quadrata di Bq

```
CLS ' pulisce lo schermo

INPUT "n=" ; n ' entra il numero dei termini delle due serie (massimo valore di n =250)

DIM A( n ), B( n ) ' dimensionamento matrici volatili per contenere i valori delle due serie di dati

FOR P = 1 TO n ' comanda richiesta caricamento matrice A(n)

LOCATE 1, 10: PRINT "A" ; P ' richiesta valori della serie An

LOCATE 2, 10: INPUT A( P ) ' caricamento matrice volatile A(n)

NEXT P ' rimanda all'istruzione FOR P... per il successivo valore della serie

FOR P = 1 TO n ' comanda richiesta caricamento matrice B(n)

LOCATE 3, 10: PRINT "B" ; P ' richiesta valori della serie Bn

LOCATE 4, 10: INPUT B( P ) ' caricamento matrice volatile B(n)

NEXT P ' rimanda alla seconda istruzione FOR P... per il successivo valore della serie

FOR i = 1 TO n ' governa il calcolo delle sommatorie progressive per il

A = A + A( i )       ' successivo calcolo   delle medie di An e Bn

B = B + B( i )

NEXT i ' rimanda a FOR i... per il completamento delle sommatorie

Am = A / n ' calcolo della media dei termini della serie An

Bm = B / n ' calcolo della media dei termini della serie Bn

FOR i = 1 TO n ' governa il calcolo delle differenze alle medie , le somme progressive dei prodotti,        &
```

```
                    ' le somme progressive dei quadrati
Ao = ( A( i ) - Am )   ' calcolo differenze alla media per A(i)
Bo = ( B( i ) - Bm )   ' calcolo differenze alla media per B(i)
SP = SP + ( Ao * Bo )  ' sommatoria progressiva dei prodotti
Aq = Aq + Ao ^ 2  ' sommatoria progressiva dei quadrati Aq
Bq = Bq + Bo ^ 2   ' sommatoria progressiva dei quadrati Bq
NEXT i   ' rimanda alla seconda istruzione FOR i...
RAq = SQR( Aq ) ' radice quadrata della sommatoria progressiva Aq
RBq = SQR( Bq )  ' radice quadrata della sommatoria progressiva Bq
C = ( SP ) / ( RAq * RBq ) ' calcolo finale del coefficiente di correlazione
DsA = RAq / SQR( n )  ' calcolo della deviazione standard per la serie An
DsB = RBq / SQR( n )  ' calcolo della deviazione standard per la serie Bn
PRINT "C=" ; C   ' presenta il coefficiente di correlazione
PRINT "Am=" ; Am   ' presenta la media della serie An
PRINT "Bm=" ; Bm   ' presenta la media della serie Bn
PRINT "DsA=" ; DsA   ' presenta il dato della deviazione standard per la serie An
PRINT "DsB=" ; DsB   ' presenta il dato della deviazione standard per la serie Bn
```

Esaminato il programma, vediamone l'impiego cercando di evidenziare come i risultati che si ottengono siano rappresentativi dei legami di interdipendenza tra le coppie di serie elaborate. A questo scopo iniziamo a considerare due serie di valori legate tra loro da un semplice rapporto di proporzionalità diretta quale può trovarsi tra il tempo di riempimento di una vasca e il volume d'acqua che in essa si deposita; dopo 7 osservazioni a distanza di 1 ora sia:

An tempo in ore	Bn volume d'acqua mc
1	.75
2	1.5
3	2.25
4	3
5	3.75
6	4.5
7	5.25

con queste due serie di dati procediamo al calcolo degli elementi elaborati dal programma:

```
                F5
n ? 7
A1
? 1
```

..... si introducono tutti i valori fino ad A7 = 7 dopo di che il programma richiede i valori di Bn
B1
? .75
...... si introducono tutti i valori fino a B7 = 5.25 , dopo l'introduzione dell'ultimo si ha la presentazione dei risultati

C = 1
Am = 4
Bm = 3
DsA = 2
DsB = 1.5

L'interpretazione del risultato relativo al coefficiente di correlazione è la seguente:
Il valore di C = 1 ci dice che le due serie di grandezze sono strettamente legate tra loro (massimo valore del coefficiente di correlazione), ciò era atteso dato che volutamente abbiamo ipotizzato un fenomeno fisico in cui l'effetto (volume d'acqua nella vasca) è proporzionale alla causa (tempo trascorso per il versamento).
In questo caso si dice che le due serie di valori sono tra loro **correlate**.
Un secondo esempio significativo si può ottenere ipotizzando che la vasca, relativa all'esercizio precedente, in cui sono stati immessi 5.25 mc d'acqua, venga svuotata completamente con regolarità con la stessa portata di immissione. Si rilevano le due serie di valori di questa nuova operazione:

An tempo in ore	Bn volume d'acqua mc
1	4.5
2	3.75
3	3
4	2.25
5	1.5
6	.75
7	0

Se i valori delle due serie vengono introdotti nel programma abbiamo:

F5
n ? 7
A1
? 1
...... si introducono i termini fino ad A7=7
B1
? 4.5
...... si introducono i termini fino a B7= 0 e si ottiene
C = -1
Am = 4
Bm = 2.25
DsA = 2
DsB = 1.5

Il valore di C = -1 ci dice che le due serie di grandezze sono strettamente legate tra loro (massimo valore del coefficiente di correlazione negativo), ciò era atteso dato che volutamente abbiamo ipotizzato un fenomeno fisico in cui l'effetto (volume d'acqua nella vasca) è dipendente dalla causa in senso decrescente (tempo trascorso per lo svuotamento).
In questo caso si dice che le due serie di valori sono **inversocorrelate**.

Un terzo esempio aiuterà a meglio comprendere l'utilità dell'algoritmo di correlazione; supponiamo che l'incremento del volume dell'acqua nella vasca non sia regolare, sia a causa di immissione discontinua, sia a causa di perdite nella vasca, e che i rilievi fatti portino a due nuove serie di valori come sotto riportato:

An tempo in ore	Bn volume d'acqua mc
1	1
2	.8
3	1.2
4	2
5	4
6	3
7	8

si avrà:

F5

n ? 7
A1
? 1
.... fino ad A 7 =7
B1
? 1
... fino a B7 = 8

C= .8540594
Am = 4
Bm = 2.857143
DsA = 2
DsB = 2.358484

Il nuovo valore di C = .8540594 ci dice che le due serie di grandezze non sono strettamente legate tra loro (il valore del coefficiente di correlazione è sensibilmente inferiore ad 1), ciò dipende dal fatto che elementi di perturbazione hanno inciso sul fenomeno in esame riducendo il rapporto di proporzionalità che era caratteristico del primo esercizio.
E' utile mostrare in quali casi il valore del coefficiente di correlazione è nullo o molto piccolo; per fare ciò si dovrebbero confrontare due serie molto numerose di valori presi casualmente, ad esempio tirando 100 volte due coppie di dadi diversamente colorati per associare al colore di una coppia i valori della serie An e al colore dell'altra coppia i valori della serie Bn.
Dato che questa operazione è tediosa si possono costruire due serie di numeri ad arte tali da ottenere che il coefficiente di correlazione tra di esse sia nullo così come mostrano le serie di valori delle seguenti tabelle:

An	Bn
4	-4
-6	-6
-2	2
4	-4

che inserite a programma permettono di ottenere:

F5

n ? 4
A l
? 4
..... fino ad A4 = 4
B l
? - 4
..... fino a B4 = - 4

C = 0
Am = 0
Bm = -3
DsA = 4.24264
DsB = 3

come era nelle intenzioni le due serie di valori An; Bn portano a C = 0, ciò denuncia la completa indipendenza di una serie rispetto all'altra. In questo caso le due serie di valori si dicono **scorrelate.**

9.2 La generazione di serie di numeri casuali

Per diversi impieghi, tra i quali le esercitazioni a scopo didattico con gli algoritmi di correlazione, sono disponibili in Qbasic alcune istruzioni che gestiscono la generazione automatica di serie di numeri casuali, tali quindi da essere, se prese con un numero sufficiente di termini, praticamente scorrelate .
Le istruzioni in oggetto sono:

RANDOMIZE TIMER
RND

L'istruzione RANDOMIZE TIMER ha il compito di. inizializzare il generatore di numeri casuali che è implementato in Qbasic.
L'istruzione RND restituisce i numeri casuali inizializzati dall'istruzione precedente, i numeri resi sono compresi tra 0 ed 1.
Vediamo come compilare un piccolo programma per la generazione di una serie di numeri casuali:

```
CLS ' pulisci lo schermo

INPUT "n" ; n ' richiesta del numero dei termini della serie di valori casuali

RANDOMIZE TIMER ' inizializza il generatore di numeri casuali

FOR i =1 TO n ' imposta la restituzione di n numeri casuali da parte dell'istruzione RND

y = RND ' restituisce gli n numeri casuali ponendoli uguali ad y

PRINT y ' presenta gli n numeri casuali

NEXT i ' rimanda all'istruzione FOR i ... per la restituzione dei successivi numeri casuali
```

Se desideriamo ad esempio una serie di 12 numeri casuali premiamo F5 e si ha:
n? 12
.
. e di seguito la sequenza di 12 numeri diversi tra loro che non si elencano dato
. che non si ritroverebbero mai uguali nel provare il programma.

E' perciò immediata la conseguenza che con due giri di programma, per lo stesso valore di n, si hanno le presentazioni di due serie di numeri diverse tra loro.
Dato che i numeri generati sono compresi tra 0 ed 1, se si vogliono valori tra 0 e 10, 0 e 100, ecc.. è necessario modificare l'istruzione y = RND in:

y = 10 * RND; y = 100 * RND; ecc...

Se si desiderano serie di numeri casuali interi si deve impiegare l'istruzione **INT**, che opportunamente inserita consente di restituire la sola parte intera di un numero:

y = INT (10 * RND); y = INT (100 * RND); . .ecc..

9.3 L'impiego del generatore di numeri casuali nei processi di correlazione

Disponendo del "generatore di numeri casuali" implementato in Qbasic è interessante vedere il comportamento dei coefficienti di correlazione quando nel programma del paragrafo 9.1 si inserisce la routine illustrata nel paragrafo 9.2.
Vediamo anzitutto come il generatore può essere inserito nel contesto del programma di calcolo citato:
Il generatore di numeri casuali prende il posto del sistema di caricamento delle memorie volatili che normalmente immagazzinano, dopo digitazione, gli n valori delle serie An e Bn; infatti per questo tipo di esercitazioni le serie An e Bn sono prodotte dal generatore stesso.
Il programma modificato in questo senso è qui compilato e commentato:

```
CLS ' pulisce lo schermo

INPUT "n=" ; n 'entra il numero dei termini delle due serie (massimo valore di n =255)

DIM A( n ), B( n ) ' dimensionamento matrici volatili per contenere i valori delle due serie di dati
                   ' prodotte dal generatore di numeri casuali

RANDOMIZE TIMER ' inizializza il generatore di numeri casuali

FOR i =1 TO n ' imposta la restituzione di numeri casuali da parte dell'istruzione RND
              ' per caricarli nelle memorie volatili A(n) ;B(n)

A(i) = RND ' restituisce n numeri casuali ponendoli nella matrice A(n)

B(i) = RND ' restituisce n numeri casuali , diversi dai precedenti, ponendoli nella matrice B(n)

NEXT i ' rimanda all'istruzione FOR i ... per la restituzione dei successivi numeri casuali

FOR i = 1 TO n ' governa il calcolo delle sommatorie progressive per il

A = A + A( i ) ' successivo calcolo delle medie di An e Bn

B = B + B( i )

NEXT i ' rimanda a FOR i... per il completamento delle sommatorie

Am = A / n ' calcolo della media dei termini della serie An

Bm = B / n ' calcolo della media dei termini della serie Bn

FOR i = 1 TO n ' governa il calcolo delle differenze alle medie , le somme progressive dei prodotti, &
```

```
                  ' le somme progressive dei quadrati
Ao = ( A( i ) - Am )  ' calcolo differenze alla media per A(i)
Bo = ( B( i ) - Bm )  ' calcolo differenze alla media per B(i)
SP = SP + ( Ao * Bo ) ' sommatoria progressiva dei prodotti
Aq = Aq + Ao ^ 2  ' sommatoria progressiva dei quadrati Aq
Bq = Bq + Bo ^ 2  ' sommatoria progressiva dei quadrati Bq
NEXT i  ' rimanda alla seconda istruzione FOR i...
RAq = SQR( Aq )  ' radice quadrata della sommatoria progressiva Aq
RBq = SQR( Bq )  ' radice quadrata della sommatoria progressiva Bq
C = ( SP ) / ( RAq * RBq )  ' calcolo finale del coefficiente di correlazione
DsA = RAq / SQR( n )  ' calcolo della deviazione standard per la serie An
DsB = RBq / SQR( n )  ' calcolo della deviazione standard per la serie Bn
PRINT "C=" ; C  ' presenta il coefficiente di correlazione
PRINT "Am=" ; Am  ' presenta la media della serie An
PRINT "Bm=" ; Bm  ' presenta la media della serie Bn
PRINT "DsA=" ; DsA  ' presenta il dato della deviazione standard per la serie An
PRINT "DsB=" ; DsB  ' presenta il dato della deviazione standard per la serie Bn
```

L'impiego di questo programma sperimentale è utile perché permette di constatare come varia il coefficiente di correlazione in dipendenza del numero dei termini che formano le due serie di numeri casuali. Infatti per pochi termini, anche se generati in modo casuale, si ha ancora correlazione tra le due serie, aumentando il numero dei termini delle serie si vede che il valore di C decresce fino ad indicare livelli di scorrelazione sensibili.

Per verificare quanto detto facciamo girare il programma per diversi valori di n e prendiamo nota di come varia il corrispondente valore di C; questa operazione viene svolta in modo automatico senza che l'operatore possa rendersi conto dei valori che compongono le due serie messe a calcolo di volta in volta.

I risultati di questo esercizio non possono essere riportati numericamente nel testo dato che non sono mai ripetibili a seguito della variabilità casuale dei numeri delle due serie che incide naturalmente sul coefficiente di correlazione. L'andamento dei risultati, di volta in volta diversi, può essere però commentato con profitto:

La prima osservazione riguarda serie di 2 o 3 termini, per un numero così limitato di valori il coefficiente di correlazione C è naturalmente molto elevato e può oscillare casualmente intorno a +1 ed a -1. E' poco probabile infatti che pochi numeri presi a caso possano mostrare caratteristiche di scorrelazione quali quelle studiate appositamente per l'ultimo esercizio del paragrafo 9.1.

Per serie da 4 a 50 termini si nota che C decresce sensibilmente oscillando tra valori ora positivi ora negativi, compresi indicativamente in una fascia che varia da .70 a .10. Ciò dimostra che le serie iniziano a manifestare segni evidenti di scorrelazione.

Per serie da 50 a 250 termini si osservano molti casi di scorrelazione quasi totale con valori di C dell'ordine di + / - .001, si presentano ancora alcuni valori di C piazzati attorno a +/- .1.
L'esercizio svolto ha dato un'idea di come vengano trattate le serie numeriche mediante gli algoritmi di correlazione, la notevole variabilità dei risultati dei calcoli dipende soltanto dalle caratteristiche della distribuzione casuale dei valori numerici delle serie.

9.4 La correlazione tra funzioni di tabella (matrice)

E' significativa la ricerca dei coefficienti di correlazione tra funzioni di tabella che, diversamente dalle serie numeriche, devono essere confrontate mediante la **correlazione di posizione**.
La correlazione di posizione è un procedimento di calcolo uguale a quello mostrato nel paragrafo 9.1 che viene ripetuto tante volte quanti sono gli elementi della matrice, in dipendenza della posizione degli elementi di una matrice rispetto agli elementi dell'altra tenuti fissi.
Per comprendere meglio quando detto vediamo un semplice esempio che interessa due funzioni di matrice espresse come segue:

Funzione A(x)	Funzione B(x)
10	7
2	9
15	6
8	5

La prima operazione di correlazione si esegue, mediante il programma illustrato nel paragrafo 9.1, sulle due funzioni di matrice come se fossero due serie di numeri e si ricava il coefficiente di correlazione di C1 = -.6624057.
Se si spostano gli elementi della funzione B(x) di un posto in modo che risulti:

Funzione A(x)	Funzione B(x)
10	9
2	6
15	5
8	7

e con questa nuova disposizione si esegue il calcolo del secondo coefficiente di correlazione.
Si ha: C2 = -.1179627.
Se si spostano ancora di un posto gli elementi della funzione B(x) si ha

Funzione A (x)	Funzione B(x)
10	6
2	5
15	7
8	9

ed il valore del terzo coefficiente di correlazione è C3 = .3901842.
Infine l'ultimo spostamento degli elementi della funzione B(x):

Funzione A(x)	Funzione B(x)
10	5
2	7
15	9
8	6

ed il valore del quarto coefficiente di correlazione è C4 = .3901842.
Con questo procedimento abbiamo calcolato, mediante quattro spostamenti, 4 valori di C che rappresentano a loro volta una funzione di tabella:

variabile di spostamento S	Funzione C(s)
0	-.6624057
1	-.1179627
2	.3901842
3	.3901842

pertanto è più corretto, in questi casi, sostituire la dizione coefficiente di correlazione con la dizione **funzione di correlazione di matrice.**
La funzione di correlazione di matrice esprime come varia il legame di interdipendenza tra due funzioni di tabella, il numero più elevato in valore assoluto che risulta dai calcoli di C(x) è il massimo di correlazione, o di inversocorrelazione, esistente tra le due funzioni; nel nostro esempio si ha per C1= -.6624057 il massimo di inversocorrelazione.
Generalmente la ricerca del massimo della funzione di correlazione di matrice viene eseguita tra funzioni di tabella costituite da molti elementi perciò risulta non praticabile il metodo di calcolo adottato nell'esempio che ne ha soltanto 4. Un interessante programma implementato in Qbasic permette il calcolo automatico per funzioni di matrice con un massimo di 127 elementi; il programma viene ora impiegato ,come esercizio, su matrici a 16 termini.
Le dimensioni della matrice volatile che sarà distinta dalla sigla B1 dipendono dal numero degli elementi, tra ordinari e spostati, pari al prodotto $n \cdot (n - 1) = 16 \cdot (16 - 1) = 240$.
Prima di compilare il programma di calcolo della funzione di correlazione di matrice è opportuno anticiparne la struttura:

1° - Le prime 10 istruzioni che seguono **CLS** sono dedicate al caricamento manuale, da parte dell'operatore, degli elementi che costituiscono le due matrici da correlare A(n) e B(n), per un massimo di 16 elementi.
2° - Con l'istruzione **FOR k = 0 TO (n - 1)**, per ciascuno degli n valori di k, si comanda l'esecuzione di uno degli n processi di correlazione tra le due funzioni di matrice.
3° - Con le due istruzioni **FOR x = 1 TO n** si comandano le n operazioni che portano al computo del primo valore C1
4°- Con le istruzioni
y = x + k
IF y > n THEN y1 = (y - n) ELSE y1 = y
si costruisce la variabile indipendente y1 che deve gestire il caricamento degli (n · (n - 1)) elementi della matrice B1(x) derivata dalla B(n) dopo (n - 1) spostamenti così come mostrato nell'esempio dell'inizio paragrafo.

Il procedimento è spiegato con il seguente esempio svolto per n = 3:
inizia la prima scansione dei 3 elementi di A(n) e B(n) eseguita per k = 0 ed x variabile da 1 ad 3 e con y1 = y = 1; 2; 3, la scansione in x legge nell'ordine gli elementi A(1);A(2);A(3), la scansione in y1 legge nell'ordine gli elementi B(1);B(2);B(3), raggiunto n = 3 il valore di k diventa 1 (vedi istruzione al punto 2°) e tutti i valori di y = x + k sono incrementati di 1; y1 = y = 2; 3; 4 ciò provoca uno spostamento di un posto nella lettura di B(n), quando y raggiunge il valore 4 > n = 3 viene posto y1 = y - n = 4 - 3 = 1 e la scansione legge gli elementi nell'ordine A(1);A(2);A(3); B(2);B(3);B(1). Il processo si ripete per k = 2 e tutti i valori di y = x + k sono incrementati di 2; y1 = y = 3; 4; 5 ciò provoca uno secondo spostamento di un posto nella lettura di B(n), quando y raggiunge il valore 4 > n = 3 il valore di y1 viene prima posto: y1= y - n = 4 - 3 = 1 e dopo per 5 > n = 3 viene posto: y1 = y - n = 5 - 3 = 2 e la scansione legge gli elementi nell'ordine A(1); A(2); A(3); B(3); B(1); B(2).
5° - Con l'istruzione B1(x) = B(y1) si comanda il caricamento della matrice B1(x), ad (n · (n-1)) elementi, con i valori ricavati tramite spostamenti dalla B(n).
6° - Seguono le istruzioni di calcolo del coefficiente di correlazione già commentate nel programma illustrato al paragrafo 9.1.
7°- Il programma di calcolo progressivo si completa con le due istruzioni NEXT x.
8°- Il programma generale si chiude con l'istruzione NEXT k che rimanda la routine all'inizio per iniziare il calcolo di un successivo valore di C.

Il programma citato è ora compilato e commentato, esso non prevede la presentazione dei calcoli relativi alle medie ed alle deviazioni standard che se necessario possono essere aggiunte dal lettore.

```
CLS ' pulisce lo schermo

INPUT "n" ; n ' entra il numero degli elementi di matrice

DIM A(16 ), B(16), B1(255) ' dimensionamento delle matrici volatili

FOR i = 1 TO n ' governa il caricamento della matrice A(n)

PRINT "A" ; i ' presenta il simbolo A(i) per l'istruzione successiva

INPUT A( i ) ' presenta il simbolo ? per l'ingresso valori di A(n)

NEXT i ' rimanda alla prima istruzione FOR i... per il caricamento dei successivi elementi matrice A(n)

FOR i = 1 TO n ' governa il caricamento della matrice B(n)

PRINT "B" ; i ' presenta il simbolo B(i) per l'istruzione successiva

INPUT B( i ) ' presenta il simbolo ? per l'ingresso valori di B(n)

NEXT i ' rimanda alla seconda istruzione FOR i... per il caricamento dei successivi elementi matrice B(n)

FOR k = 0 TO (n - 1) ' istruzione per la formazione di una matrice B1(x)
    ' contiene tutte le possibili configurazioni di posizione degli elementi di B(n) pari ad ( n * ( n-1 ) ) elementi

FOR x = 1 TO n ' esplora la matrice A(n) da 1 ad n per (n-1) volte

y = x + k ' si forma la variabile per la matrice B1(x)

IF y > n THEN y1 = ( y - n ) ELSE y1 = y          &
```

```
B1( x ) = B( y1 )  ' si caricano gli n valori di B(n) disposti in ( n * ( n-1 ) ) posizioni d'ordine diverse in B1(x)

A = A + A( x )  ' calcolo medie

B = B + B1( x )

NEXT x

Am = A / n   ' calcolo della medie Am dopo le sommatorie

Bm = B / n   ' calcolo della media Bm dopo le sommatorie

FOR  x = 1  TO  n

Ao = ( A( x ) - Am )   ' calcolo differenze alla media per A(x)

Bo = ( B1( x ) - Bm )  ' calcolo differenze alla media per B1(x)

SP = SP + ( Ao * Bo )  ' sommatoria progressiva dei prodotti

Aq = Aq + Ao ^ 2  ' sommatoria progressiva dei quadrati Aq

Bq = Bq + Bo ^ 2   ' sommatoria progressiva dei quadrati Bq

NEXT x

RAq = SQR( Aq )  ' radice quadrata della sommatoria progressiva Aq

RBq = SQR( Bq )  ' radice quadrata della sommatoria progressiva Bq

C = ( SP ) / ( RAq * RBq )  ' calcolo finale degli n coefficienti di correlazione

PRINT "C";  k  ; "=" ; C  ' presentazione dei coefficienti di correlazione.

A = 0    ' azzeramento memorie per i successivi calcoli di C
B = 0
Am = 0
Bm = 0
Ao = 0
Bo = 0
SP = 0
Aq = 0
Bq = 0

NEXT k
```

Per provare il programma proponiamoci la determinazione della funzione di correlazione di matrice delle due funzioni di tabella a 10 elementi (n = 10):

variabile	Funzione	Funzione
x	A(x)	B(x)
1	1.7	2.5
2	3.4	2.9
3	5.0	3.2
4	6.4	3.5
5	7.6	3.0
6	8.6	3.7

7	9.3	.6
8	9.8	1.0
9	10.0	1.7
10	11.0	2.0

F5

n? 10
A1
? 1.7
A2
?
3.4
.... di seguito fino a
A10
? 11
B1
? 2.5
B2
? 2.9
..... di seguito fino a
B10
? 2 dopo l'introduzione dell'ultimo elemento di B(x) si ha

C0 = -. 4759656
C1 = -. 5531418
C2 = -. 4766486
C3 = -. 2539985
C4 = . 1339315
C5 = . 4026823
C6 = . 989358
C7 = . 4487832
C8 = 3. 011922 E-02
C9 = -. 2451199

Il calcolo ha portato alla determinazione della funzione di correlazione di matrice che presenta il massimo di correlazione per C6 = .989358, è utile in molte applicazioni evidenziare, sia il valore massimo di correlazione, sia il numero (S) corrispondente allo spostamento effettuato sugli elementi di B(x) per ottenere tale massimo; nel nostro caso abbiamo S = 6.

La funzione di correlazione di matrice viene così espressa:

variabile di spostamento S	C(S)
0	-.4759656
1	-.5531418
2	-.4766486
3	-.2539985
4	.1339315
5	.4026823
6	.989358
7	.4487832
8	3.01192 E-02
9	-.2451199

dove il valore C(0) rappresenta la correlazione esistente tra A(x) e B(x) eseguita senza alcun spostamento degli elementi di B(x),......, dove il valore di C(9) rappresenta la correlazione esistente tra le due funzioni di matrice dopo aver eseguito il nono spostamento degli elementi di B(x). E' di fondamentale importanza rimarcare che le operazioni di correlazioni tra le due funzioni sono eseguite nel presupposto che **A(x) rimanga sempre invariata**, mentre gli spostamenti avvengono soltanto negli elementi di B(x).

9.5 La correlazione tra fenomeni ondulatori casuali

Si definiscono fenomeni ondulatori casuali quelli esprimibili mediante funzioni f(t) che mutano nel tempo sia in ampiezza che in polarità in modo casuale.
Gli algoritmi che abbiamo illustrato nel paragrafo 9.1 riguardano la ricerca dei valori dei coefficienti di correlazione tra serie di grandezze in numero discreto, questi metodi di calcolo non si adattano alla determinazione degli omologhi indicatori della correlazione tra funzioni che esprimono dei fenomeni ondulatori casuali.
Una certa analogia esiste invece tra la correlazione di funzioni di matrice e funzioni ondulatorie casuali, queste ultime possono considerarsi "estensioni" delle prime con un numero infinito di elementi nelle quali lo spostamento (S), definito con un numero discreto di valori, viene sostituito da una variabile di spostamento continua (r).
Gli indicatori del grado di correlazione tra fenomeni ondulatori casuali hanno caratteristiche di vere e proprie funzioni matematiche, per tale motivo sono detti **funzioni di correlazione**.
La teoria sulle funzioni di correlazione è molto vasta e non può essere trattata in questa sede, sede che si propone soltanto di fornire delle informazioni sui metodi di calcolo implementabili in Qbasic.
Le funzioni di correlazione si ottengono per via analitica mediante complesse elaborazioni, i loro algoritmi sono però facilmente implementabile in Qbasic come ora dimostreremo.

9.5.1 La funzione di correlazione tra due fenomeni ondulatori casuali in banda (0 - F)

Se due fenomeni casuali, definiti rispettivamente dalle funzioni del tempo f1(t) e f2(t), hanno tutte le loro componenti ondulatorie contenute in un intervallo di frequenza compreso tra la frequenza zero e la frequenza F, e se in tale intervallo l'ampiezza delle componenti dei due fenomeni è mediamente uniforme, la funzione di correlazione normalizzata tra i fenomeni stessi è data da:

$$C(r) = \frac{\mathrm{Sen}\,[\,2\pi F\,(r - rf)\,]}{[\,2\pi F\,(r - rf)\,]}$$

dove

C(r) = simbolo della funzione di correlazione

F = frequenza estrema dell'intervallo

r = variabile indipendente di spostamento temporale di f2(t) rispetto a f1(t)

rf = valore iniziale e fisso di spostamento temporale di f1(t) rispetto a f2(t)

Si deve osservare che in alcuni casi può essere rf = 0, ciò significa che le due funzioni sono correlate per r = 0.
Un esempio fisico è utile per inquadrare meglio questa nuova funzione:

Sia f(t) un fenomeno ondulatorio casuale dovuto a vibrazioni meccaniche le cui oscillazioni sono contenute nell'intervallo compreso tra 0 e 1000 Hz . L'azione vibratoria di f(t) si propaghi su due percorsi diversi, percorso 1 e percorso 2, con il percorso 1 superiore al percorso 2, per poi essere percepita nello stesso punto d'ascolto A.
Indicando con f1(t) la vibrazione che si propaga lungo il percorso 1 supponiamo che questo sia coperto in .0005 Sec.
Indicando con f2(t) la vibrazione che si propaga lungo il percorso 2 supponiamo che questo sia coperto in .0003 Sec.
E' evidente che la vibrazione portata in A da f1(t) sarà ritardata rispetto alla vibrazione portata in A da f2(t) del tempo rf = .0005 - .0003 = .0002 Sec.
La funzione di correlazione tra i due fenomeni vibratori presenti nel punto A, funzione che esprime il legame di interdipendenza tra f1(t) e f2(t), sarà pertanto:

$$C(r) = \frac{\mathrm{Sen}\,[\,2\,\pi\,1000\,(\,r - .0002\,)\,]}{[\,2\,\pi\,1000\,(\,r - .0002\,)\,]}$$

A questo punto non resta che computare l'espressione ottenuta, in funzione della variabile di posizione temporale (r), per ottenere l'andamento della funzione di correlazione interessata.
Il computo ed il grafico di C(r) si ottengono dal programma generale che si avvale della corrispondenza simbolica in Qbasic della funzione C(r):

C(r) = SIN (2 * 3.14 * F * (r - rf)) / (2 * 3.14 * F * (r - rf))

che deve essere sviluppata per tutti i valori di r compresi nell'intervallo 0 - ro dove ro >> rf
Il programma compilato e commentato è:

```
' Iniziare digitando la routine per il grafico del reticolo

LINE ( 0 , 160 ) - ( 460 , 160 )  ' comanda il tracciamento dell'asse delle ascisse (asse X)
                                  ' per coordinate a 2 quadranti

LINE ( 0 , 0 ) - ( 0 , 320 )  ' comanda il tracciamento dell'asse delle ordinate ( asse Y)
                              ' per coordinate a 2 quadranti

LOCATE 8,66 : PRINT "F"   ' presentazione simbolo F per istruzione successiva

LOCATE 9,66 :  INPUT F     ' presentazione simbolo  ?  per introduzione valore di F

LOCATE 10,66 : PRINT "rf "  ' presentazione simbolo  rf  per istruzione successiva

LOCATE 11,66 :  INPUT rf    ' presentazione simbolo  ?  per introduzione valore di rf

LOCATE 12,66 :  PRINT "ro"   ' presentazione simbolo ro per istruzione successiva

LOCATE 13,66 : INPUT ro   ' presentazione simbolo  ?  per introduzione valore di ro

FOR  r = .0000001  TO  ro  STEP  ( ro / 1000 )  ' impostazione campo  variabilità di r  a 1000 passi fissi

C( r ) = SIN ( 2 * 3.14 * F * ( r - rf ) ) / ( 2 * 3.14 * F * ( r - rf ) )  ' calcolo C(r)

PSET( 460 * r / ro , 160 - 160 * C ( r ) ),14  ' tracciamento del grafico andamento C(r) con      &
```

' normalizzazione automatica asse delle ascisse

NEXT r ' rimanda all'istruzione FOR r... per il calcolo dei successivi punti di C(r)

L'esempio fisico illustrato può essere concluso con l'applicazione del programma ora compilato assumendo per ro il valore .002 >> rf; si preme F5 e si ha:

F
? 1000
rf
? .0002
ro
? .002

dopo l'introduzione del valore di ro si ha la presentazione, in giallo, della curva di correlazione riportata in figura 45 in cui l'asse delle ordinate è diviso in 20 intervalli da .1 e l'asse delle ascisse in 20 intervalli da .0001 Sec.
Come avevamo anticipato si constata che il legame di correlazione tra due funzioni è a sua volta una funzione e come tale è rappresentabile graficamente per un qualsiasi numero di valori compatibilmente con il numero dei pixel disponibili sullo schermo.
Dalla curva si osserva:
-La C(r) presenta il suo valore massimo in corrispondenza di r = .0002 Sec. così come era prevedibile dato che tanto è il ritardo che la f1(t) ha rispetto alla f2(t).
-Il profilo della C(r) ha la nota caratteristica delle funzioni Sen x /x.
-La curva di C(r) presenta il primo zero per un valore di
r = ro + 1 / 2F = .0002 + 1 / (2 · 1000) = .0007 Sec.
Ciò indica che per tale valore le due funzioni f1(t) e f2(t) sono completamente scorrelate, cioè il loro legame di interdipendenza è nullo. La cosa può stupire dato che abbiamo ipotizzato che le due vibrazioni provengano dallo stesso fenomeno ondulatorio. La realtà fisica dell'esempio conferma invece questo comportamento che è inoltre mostrato dall'andamento della C(r) che, al di là di altri valori di zero netto, tende a decrescere d'ampiezza indefinitamente per valori di r crescenti indefinitamente.



Figura 45
Funzione di correlazione
banda 0 - 1000 Hz
rf = .0002 Sec.

9.5.2 La funzione di correlazione tra due fenomeni ondulatori casuali in banda (F1 - F2)

Se due fenomeni casuali, definiti rispettivamente dalle funzioni del tempo f1(t) e f2(t), hanno tutte le loro componenti ondulatorie contenute in un intervallo di frequenza compreso tra la frequenza F1 e la frequenza F2 e se in tale intervallo l'ampiezza media delle componenti dei due fenomeni è uniforme, la funzione di correlazione normalizzata tra i fenomeni stessi è data da:

$$C(r) = \frac{\text{Sen}\,[\,2\,\pi\ DF\ (\,r - rf\,)\,]}{[\,2\,\pi\ DF\ (\,r - rf\,)\,]}\ \text{Cos}\,[\,2\,\pi\, Fo\,(\,r - rf\,)\,]$$

dove

C(r) = simbolo della funzione di correlazione
DF = (F2 - F1) / 2
Fo = (F2 + F1) / 2
r = variabile indipendente di spostamento temporale di f2(t) rispetto a f1(t)
rf = valore iniziale e fisso di spostamento temporale di f1(t) rispetto a f2(t)

L ' esempio fisico riportato nel paragrafo 9.5.1 si adatta naturalmente anche a questo tipo di fenomeni casuali per cui possiamo scrivere:
Sia f(t) un fenomeno casuale dovuto a vibrazioni meccaniche le cui oscillazioni sono contenute nell'intervallo compreso tra 5000 Hz e 10000 Hz. Indicando con f1(t) la vibrazione che si propaga lungo il percorso 1 supponiamo che questo sia coperto in .0001 Sec.
Indicando con f2(t) la vibrazione che si propaga lungo il percorso 2 supponiamo che questo sia coperto in .00006 Sec. La vibrazione portata in A da f1(t) sarà ritardata rispetto alla vibrazione portata in A da f2(t) del tempo rf = .0001 - .00006 = .00004 Sec.
Per l'impostazione della funzione di correlazione tra i due fenomeni vibratori presenti nel punto A, funzione che esprime il legame di interdipendenza tra f1(t) e f2(t), si deve anzitutto computare:

$$DF = (\ F2 - F1\ ) / 2\ =\ (\ 10000 - 5000\ ) / 2 = 2500\ \text{Hz}$$

$$Fo = (\ F2 + F1\ ) / 2\ =\ (\ 10000\ + 5000\ ) / 2 = 7500\ \text{Hz}$$

da cui si ha

$$C(r) = \frac{\text{Sen}\,[\,2\,\pi\ 2500\ (\,r - .00004\,)\,]}{[\,2\,\pi\ 2500\ (\,r - .00004\,)\,]}\ \text{Cos}\,[\,2\,\pi\ 7500\,(\,r - .00004\,)\,]$$

Il calcolo ed il grafico di C(r) si ottengono dal programma generale che si avvale della corrispondenza simbolica in Qbasic della funzione C(r) spezzata per comodità di compilazione in due parti:

```
Y1= SIN ( 2 * 3.14 * ( ( F2 - F1 ) / 2 ) * ( r - rf ) ) / ( 2 * 3.14 * ( ( F2 - F1 ) / 2) * ( r - rf ) )

Y2 = COS ( 2 * 3.14 * ( ( F2 + F1 ) / 2 ) * ( r - rf ) )

C( r ) = Y1 * Y2
```

che deve essere sviluppata per tutti i valori di r compresi nell'intervallo 0-ro dove ro >> rf
Il programma compilato e commentato è:

```
'Iniziare digitando la routine per il grafico del reticolo

LINE (0 , 160 ) - ( 460 , 160 ) ' comanda il tracciamento dell'asse delle ascisse (asse X)
                                ' per coordinate a 2 quadranti
LINE ( 0 , 0 ) - ( 0 , 320 )  ' comanda il tracciamento dell'asse delle ordinate ( asse Y)
                              ' per coordinate a 2 quadranti

LOCATE 8,66 : PRINT "F1"  ' presentazione simbolo F1 per istruzione successiva

LOCATE 9,66 : INPUT F1     ' presentazione simbolo ? per introduzione valore di F1

LOCATE 10,66 : PRINT "F2"  ' presentazione simbolo F2 per istruzione successiva

LOCATE 11,66 : INPUT F2  ' presentazione simbolo ? per introduzione valore di F2

LOCATE 12,66 : PRINT "rf " ' presentazione simbolo rf per istruzione successiva

LOCATE 13,66 : INPUT rf   ' presentazione simbolo ? per introduzione valore di rf

LOCATE 14,66 : PRINT "ro"  ' presentazione simbolo ro per istruzione successiva

LOCATE 15,66 : INPUT ro  ' presentazione simbolo ? per introduzione valore di ro

FOR r = .0000001 TO ro STEP ( ro / 1000 ) ' impostazione campo variabilità di r a 1000 passi fissi

Y1= SIN ( 2 * 3.14 * ( ( F2 - F1 ) / 2 ) * ( r - rf ) ) / ( 2 * 3.14 * ( ( F2 - F1 ) / 2 ) * ( r - rf ) )

Y2 = COS ( 2 * 3.14 * ( ( F2 + F1 ) / 2 ) * ( r - rf ) )

C( r ) = Y1 * Y2

PSET( 460 * r / ro , 160 - 160 * C ( r ) ),14 ' tracciamento del grafico andamento C(r) con
                                              ' normalizzazione automatica asse delle ascisse

NEXT r ' rimanda all'istruzione FOR r... per il calcolo dei successivi punti di C(r)
```

L'esempio può essere concluso con l'applicazione del programma ora compilato assumendo per ro il valore .0004 >> rf:

```
F5
                                F1
                                ? 5000
                                F2
                                ? 10000
                                rf
                                ? . 00004
                                ro
                                ? . 0004
```

dopo l'introduzione del valore di ro si ha la presentazione, in giallo, della curva di correlazione riportata in figura 46 in cui l'asse delle ordinate è diviso in 20 intervalli da .1 e l'asse delle ascisse in 20 intervalli da .00002 Sec.

Dalla curva si osserva:

-La C(r) presenta il suo valore massimo in corrispondenza di r = .00004 Sec., così come era prevedibile dato che tanto è il ritardo che la f1(t) ha rispetto alla f2(t).
-Il profilo della C(r) ondula secondo la funzione coseno modulata in Sen x/x, infatti la funzione di correlazione è il prodotto di due funzioni: una cosinusoidale, che varia rapidamente con il variare di r, e una del tipo Sen x/x, che varia, con il variare di r, meno rapidamente della prima.
-Il legame di correlazione di questo tipo di fenomeni dipende prevalentemente dalla larghezza della banda delle frequenze che li compongono e relativamente dal valore di r.



Figura 46
Funzione di correlazione
in banda 5000 - 10000 Hz
rf = .00004 Sec.

9.6 Le funzioni di correlazione per fenomeni casuali a due stati

Per concludere questo capitolo è necessario accennare a due utilissimi algoritmi di correlazione che giuocano un ruolo fondamentale nell'analisi dei fenomeni casuali.
I fenomeni ondulatori casuali, dei tipi già trattati in precedenza, sono definiti da una funzione f(t) che muta nel tempo in ampiezza e polarità in modo casuale.
Un fenomeno casuale a due stati è definito invece da una funzione f (t) che ha sempre ampiezza costante e varia soltanto in polarità in modo casuale.

9.6.1 La funzione di correlazione tra fenomeni casuali a due stati in banda (0 - F)

Se due fenomeni casuali a due stati, definiti rispettivamente dalle funzioni del tempo f1(t) e f2(t), hanno tutte le loro componenti frequenziali contenute in un intervallo di frequenza compreso tra la frequenza zero e la frequenza F, la funzione di correlazione normalizzata tra i fenomeni stessi è data da:

$$Cl(r) = (2/\pi)\ \mathrm{Arcsen}\ \frac{\mathrm{Sen}\,[\,2\pi F\ (\,r - rf\,)\,]}{[\,2\pi F\ (\,r - rf\,)\,]}$$

dove

Cl(r) = simbolo della funzione di correlazione tra funzioni casuali a due stati
F = frequenza estrema dell'intervallo
r = variabile indipendente di spostamento temporale di f2(t) rispetto a f1(t)

rf = valore iniziale e fisso di spostamento temporale di f1(t) rispetto a f2(t)

Il computo ed il grafico di Cl(r) si ottengono dal programma generale che si avvale della corrispondenza simbolica in Qbasic della funzione Cl(r) che per semplificare la compilazione e scritta con due istruzioni:

Y = SIN (2 * 3.14 * F * (r - rf)) / (2 * 3.14 * F * (r - rf))

Cl(r) = (2 / 3.14) * ATN (Y / SQR (-Y * Y + 1))

che deve essere sviluppata per tutti i valori di r compresi nell'intervallo 0-ro dove ro >> rf.
Il programma compilato e commentato è il seguente:

```
' Iniziare digitando la routine per il grafico del reticolo

LINE ( 0 , 160 ) - ( 460 , 160 ) ' comanda il tracciamento dell'asse delle ascisse (asse X)
                                 ' per coordinate a 2 quadranti

LINE ( 0 , 0 ) - ( 0 , 320 )  ' comanda il tracciamento dell'asse delle ordinate ( asse Y)
                              ' per coordinate a 2 quadranti

LOCATE 8,66 : PRINT "F"    ' presentazione simbolo F per istruzione successiva

LOCATE 9,66 : INPUT F   ' presentazione simbolo ? per introduzione valore di F

LOCATE 10,66 : PRINT "rf "  ' presentazione simbolo rf per istruzione successiva

LOCATE 11,66 : INPUT rf   ' presentazione simbolo ? per introduzione valore di rf

LOCATE 12,66 : PRINT "ro"  ' presentazione simbolo ro per istruzione successiva

LOCATE 13,66 : INPUT ro   ' presentazione simbolo ? per introduzione valore di ro

FOR r = .0000001 TO ro STEP ( ro / 1000 )  ' impostazione campo variabilità di r a 1000 passi fissi

Y = SIN ( 2 * 3.14 * F * ( r - rf ) ) / ( 2 * 3.14 * F * ( r - rf ) )  ' calcolo di Cl(r)

Cl( r ) = ( 2 / 3.14 ) * ATN ( Y / SQR ( -Y * Y + 1 ) )           ' in due passi di programma

PSET( 460 * r / ro , 160 - 160 * Cl ( r ) ),14 ' tracciamento del grafico andamento Cl(r) con
                                               ' normalizzazione automatica asse delle ascisse
NEXT r ' rimanda all'istruzione FOR r... per il calcolo dei successivi punti di Cl(r)
```

Se proviamo il programma ora compilato assumendo gli stessi valori utilizzati nell'esercizio del paragrafo 9.5.1 otteniamo

F5

F
? 1000
rf
? .0002
ro
? .002

dopo l'introduzione del valore di ro si ha la presentazione, in giallo, della curva di correlazione riportata in figura 47 in cui l'asse delle ordinate è diviso in 20 intervalli da .1 e l'asse delle ascisse in 20 intervalli da .0001 Sec.



Figura 47
Funzione di correlazione per fenomeni a due stati in banda 0 - 1000 Hz
rf = .0002 Sec.

Dalla curva si osserva:

-Il profilo della Cl(r) mostra una cuspide in corrispondenza del massimo, il restante andamento segue di massima la funzione Sen x/x.
-La Cl(r) presenta il suo valore massimo in corrispondenza di r = .0002 Sec.
-La curva di Cl(r) presenta il primo zero per un valore di r = ro +1/ 2F = .0002 + 1 / (2 ·1000) = = .0007 Sec.

9.6.2 La funzione di correlazione tra due fenomeni casuali a due stati in banda (F1 - F2).

Se due fenomeni casuali a due stati, definiti rispettivamente dalle funzioni del tempo f1(t) e f2(t), hanno tutte le loro componenti frequenziali contenute in un intervallo di frequenza compreso tra la frequenza F1 e la frequenza F2, la funzione di correlazione normalizzata tra i fenomeni stessi è data da:

$$Cl(r) = (2/\pi)\,\mathrm{Arcsen}\left\{ \frac{\mathrm{Sen}\,[\,2\pi\ DF\ (r - rf)\,]}{[\,2\pi\ DF\ (r - rf)\,]}\ \mathrm{Cos}\,[\,2\pi\,Fo\,(r - rf)\,] \right\}$$

dove

Cl(r) = simbolo della funzione di correlazione tra funzioni casuali a due stati
DF = (F2 - F1) / 2
Fo = (F2 + F1) / 2
r = variabile indipendente di spostamento temporale di f2(t) rispetto a f1(t)
rf = valore iniziale e fisso di spostamento temporale di f1(t) rispetto a f2(t)

Il computo ed il grafico di Cl(r) si ottengono dal programma generale che si avvale della corrispondenza simbolica in Qbasic della funzione Cl(r) spezzata per comodità di compilazione in quattro parti:

Y1 = SIN (2 * 3.14 * ((F2 - F1) / 2) * (r - rf)) / (2 * 3.14 * ((F2 - F1) / 2) * (r - rf))

Y2 = COS (2 * 3.14 * ((F2 + F1) / 2)* (r - rf))

Y3 = Y1 * Y2

Cl(r) = (2 / 3.14) * ATN (Y3 / SQR (-Y3 * Y3 + 1))

che deve essere sviluppata per tutti i valori di r compresi nell'intervallo 0-ro dove ro >> rf
Il programma compilato e commentato è il seguente:

```
' Iniziare digitando la routine per il grafico del reticolo

LINE ( 0 , 160 ) - ( 460 , 160 )  ' comanda il tracciamento dell'asse delle ascisse (asse X)
                                  ' per coordinate a 2 quadranti

LINE ( 0 , 0 ) - ( 0 , 320 )  ' comanda il tracciamento dell'asse delle ordinate ( asse Y)
                              ' per coordinate a 2 quadranti

LOCATE 8,66 : PRINT "F1"  ' presentazione simbolo F1 per istruzione successiva

LOCATE 9,66 : INPUT F1  ' presentazione simbolo ? per introduzione valore di F1

LOCATE 10,66 : PRINT "F2"  ' presentazione simbolo F2 per istruzione successiva

LOCATE 11,66 : INPUT F2  ' presentazione simbolo ? per introduzione valore di F2

LOCATE 12,66 : PRINT "rf "  ' presentazione simbolo rf per istruzione successiva

LOCATE 13,66 : INPUT rf  ' presentazione simbolo ? per introduzione valore di rf

LOCATE 14,66 : PRINT "ro"  ' presentazione simbolo ro per istruzione successiva

LOCATE 15,66 : INPUT ro  ' presentazione simbolo ? per introduzione valore di ro

FOR r = .0000001 TO ro STEP ( ro / 1000 ) ' impostazione campo variabilità di r a 1000 passi fissi  &
```

```
Y1= SIN ( 2 * 3.14 * ( ( F2 - F1 ) / 2) * ( r - rf ) ) / ( 2 * 3.14 * ( ( F2 - F1 ) / 2) * ( r - rf ) )

Y2 = COS ( 2 * 3.14 * ( ( F2 + F1 ) / 2 ) * ( r - rf ) )   ' calcolo della Cl(r) in 4 passi di programma

Y3 = Y1 * Y2

Cl (r ) = ( 2 / 3.14 ) * ATN ( Y3 / SQR ( -Y3 * Y3 + 1 ) )

PSET( 460 * r / ro , 160 - 160 * Cl( r ) ),14   ' tracciamento del grafico andamento Cl(r) con
                                                ' normalizzazione automatica asse delle ascisse

NEXT r  ' rimanda all'istruzione FOR r... per il calcolo dei successivi punti di Cl(r)
```

Si prova il programma con i dati impiegati nell'esercizio del paragrafo 9.5.2

```
F5
                    F1
                    ? 5000
                    F2
                    ? 10000
                    rf
                    ? .00004
                    ro
                    ? .0004
```

dopo l'introduzione del valore di ro si ha la presentazione della curva di figura 48



Figura 48
Funzione di correlazione per fenomeni a due stati banda 5000 - 10000 Hz rf = .00004 Sec.

Dalla figura si osserva:
-l'asse delle ordinate è diviso in 20 intervalli da .1
-l'asse delle ascisse è diviso in 20 intervalli da .00002 Sec.
-Il profilo della Cl(r) presenta una cuspide sul massimo, lontano dal massimo la curva ondula secondo la funzione coseno modulata in Sen x/x .
-La Cl(r) presenta il suo valore massimo in corrispondenza di r = .00004 Sec.

-Il legame di correlazione di questo tipo di fenomeni dipende prevalentemente dalla larghezza della banda delle frequenze che li compongono e relativamente dal valore di r.